# NAG Library Routine Document

# F07CNF (ZGTSV)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

## 1 Purpose

F07CNF (ZGTSV) computes the solution to a complex system of linear equations

$$AX = B,$$

where $A$ is an $n$ by $n$ tridiagonal matrix and $X$ and $B$ are $n$ by $r$ matrices.

## 2 Specification

```
SUBROUTINE F07CNF (N, NRHS, DL, D, DU, B, LDB, INFO)

INTEGER                N, NRHS, LDB, INFO
COMPLEX (KIND=nag_wp)  DL(*), D(*), DU(*), B(LDB,*)
```

The routine may be called by its LAPACK name ***zgtsv***.

## 3 Description

F07CNF (ZGTSV) uses Gaussian elimination with partial pivoting and row interchanges to solve the equations $AX = B$. The matrix $A$ is factorized as $A = PLU$, where $P$ is a permutation matrix, $L$ is unit lower triangular with at most one nonzero subdiagonal element per column, and $U$ is an upper triangular band matrix, with two superdiagonals.

Note that the equations $A^{\mathrm{T}}X = B$ may be solved by interchanging the order of the arguments DU and DL.

## 4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia http://www.netlib.org/lapack/lug

## 5 Parameters

1: N – INTEGER *Input*

*On entry*: $n$, the number of linear equations, i.e., the order of the matrix $A$.

*Constraint*: $\mathrm{N} \geq 0$.

2: NRHS – INTEGER *Input*

*On entry*: $r$, the number of right-hand sides, i.e., the number of columns of the matrix $B$.

*Constraint*: $\mathrm{NRHS} \geq 0$.

3: DL($*$) – COMPLEX (KIND=nag_wp) array *Input/Output*

**Note**: the dimension of the array DL must be at least $\max(1, \mathrm{N} - 1)$.

*On entry*: must contain the $(n - 1)$ subdiagonal elements of the matrix $A$.

*On exit*: if no constrains are violated, DL is overwritten by the $(n - 2)$ elements of the second superdiagonal of the upper triangular matrix $U$ from the $LU$ factorization of $A$, in $\mathrm{DL}(1), \mathrm{DL}(2), \ldots, \mathrm{DL}(n - 2)$.

4: D($*$) – COMPLEX (KIND=nag_wp) array *Input/Output*

**Note**: the dimension of the array D must be at least $\max(1, \text{N})$.

*On entry*: must contain the $n$ diagonal elements of the matrix $A$.

*On exit*: if no constraints are violated, D is overwritten by the $n$ diagonal elements of the upper triangular matrix $U$ from the $LU$ factorization of $A$.

5: DU($*$) – COMPLEX (KIND=nag_wp) array *Input/Output*

**Note**: the dimension of the array DU must be at least $\max(1, \text{N} - 1)$.

*On entry*: must contain the $(n - 1)$ superdiagonal elements of the matrix $A$.

*On exit*: if no constraints are violated, DU is overwritten by the $(n - 1)$ elements of the first superdiagonal of $U$.

6: B(LDB,$*$) – COMPLEX (KIND=nag_wp) array *Input/Output*

**Note**: the second dimension of the array B must be at least $\max(1, \text{NRHS})$.

**Note**: To solve the equations $Ax = b$, where $b$ is a single right-hand side, B may be supplied as a one-dimensional array with length $\text{LDB} = \max(1, \text{N})$.

*On entry*: the $n$ by $r$ right-hand side matrix $B$.

*On exit*: if $\text{INFO} = 0$, the $n$ by $r$ solution matrix $X$.

7: LDB – INTEGER *Input*

*On entry*: the first dimension of the array B as declared in the (sub)program from which F07CNF (ZGTSV) is called.

*Constraint*: $\text{LDB} \geq \max(1, \text{N})$.

8: INFO – INTEGER *Output*

*On exit*: $\text{INFO} = 0$ unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

Errors or warnings detected by the routine:

$\text{INFO} < 0$

If $\text{INFO} = -i$, the $i$th argument had an illegal value. An explanatory message is output, and execution of the program is terminated.

$\text{INFO} > 0$

If $\text{INFO} = i$, $u_{ii}$ is exactly zero, and the solution has not been computed. The factorization has not been completed unless $i = \text{N}$.

## 7 Accuracy

The computed solution for a single right-hand side, $\hat{x}$, satisfies an equation of the form

$$(A + E)\hat{x} = b,$$

where

$$\|E\|_1 = O(\epsilon)\|A\|_1$$

and $\epsilon$ is the ***machine precision***. An approximate error bound for the computed solution is given by

$$\frac{\|\hat{x} - x\|_1}{\|x\|_1} \leq \kappa(A)\frac{\|E\|_1}{\|A\|_1},$$

where $\kappa(A) = \left\| A^{-1} \right\|_1 \|A\|_1$, the condition number of $A$ with respect to the solution of the linear equations. See Section 4.4 of Anderson *et al.* (1999) for further details.

Alternatives to F07CNF (ZGTSV), which return condition and error estimates are F04CCF and F07CPF (ZGTSVX).

# 8 Further Comments

The total number of floating point operations required to solve the equations $AX = B$ is proportional to $nr$.

The real analogue of this routine is F07CAF (DGTSV).

# 9 Example

This example solves the equations

$$Ax = b,$$

where $A$ is the tridiagonal matrix

$$A = \begin{pmatrix} -1.3 + 1.3i & 2.0 - 1.0i & 0 & 0 & 0 \\ 1.0 - 2.0i & -1.3 + 1.3i & 2.0 + 1.0i & 0 & 0 \\ 0 & 1.0 + 1.0i & -1.3 + 3.3i & -1.0 + 1.0i & 0 \\ 0 & 0 & 2.0 - 3.0i & -0.3 + 4.3i & 1.0 - 1.0i \\ 0 & 0 & 0 & 1.0 + 1.0i & -3.3 + 1.3i \end{pmatrix}$$

and

$$b = \begin{pmatrix} 2.4 - 5.0i \\ 3.4 + 18.2i \\ -14.7 + 9.7i \\ 31.9 - 7.7i \\ -1.0 + 1.6i \end{pmatrix}.$$

## 9.1 Program Text

```
    Program f07cnfe

!     F07CNF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
      Use nag_library, Only: nag_wp, zgtsv
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Integer                          :: info, n
!     .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: b(:), d(:), dl(:), du(:)
!     .. Executable Statements ..
      Write (nout,*) 'F07CNF Example Program Results'
      Write (nout,*)
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) n

      Allocate (b(n),d(n),dl(n-1),du(n-1))

!     Read the tridiagonal matrix A and the right hand side B from
!     data file

      Read (nin,*) du(1:n-1)
```

```
      Read (nin,*) d(1:n)
      Read (nin,*) dl(1:n-1)
      Read (nin,*) b(1:n)

!     Solve the equations Ax = b for x
!     The NAG name equivalent of zgtsv is f07cnf
      Call zgtsv(n,1,dl,d,du,b,n,info)

      If (info==0) Then

!       Print solution

        Write (nout,*) 'Solution'
        Write (nout,99999) b(1:n)

      Else
        Write (nout,99998) 'The (', info, ',', info, ')', &
          ' element of the factor U is zero'
      End If

99999 Format (4(' (',F8.4,',',F8.4,')':))
99998 Format (1X,A,I3,A,I3,A,A)
    End Program f07cnfe
```

## 9.2 Program Data

```
F07CNF Example Program Data
   5                                                   :Value of N
 (  2.0, -1.0) (  2.0,  1.0) ( -1.0,  1.0) (  1.0, -1.0) :End of DU
 ( -1.3,  1.3) ( -1.3,  1.3) ( -1.3,  3.3) ( -0.3,  4.3)
 ( -3.3,  1.3)                                           :End of D
 (  1.0, -2.0) (  1.0 , 1.0) (  2.0, -3.0) (  1.0,  1.0) :End of  DL
 (  2.4, -5.0) (  3.4, 18.2) (-14.7,  9.7) ( 31.9, -7.7)
 ( -1.0,  1.6)                                           :End of B
```

## 9.3 Program Results

```
 F07CNF Example Program Results

 Solution
 (  1.0000,  1.0000) (  3.0000, -1.0000) (  4.0000,  5.0000) ( -1.0000, -2.0000)
 (  1.0000, -1.0000)
```